It might be a good idea to evaluate the entire notebook before reading it. You can do it from the keyboard: Alt+v o, or from the go to the Menu item Evaluation, then: Evaluate Notebook.

In[1]:=
```
Plot3D[x y, {x, 0, 1}, {y, 0, 1}]
```

Out[1]=



In[2]:=
```
? RegionPlot
```

RegionPlot[*pred*, {*x*, x_{min}, x_{max}}, {*y*, y_{min}, y_{max}}] makes a plot showing the region in which *pred* is True. ≫

In[3]:=
```
RegionPlot[And[0 < x, x < 1, 0 < y, y < 1 - x], {x, 0, 1}, {y, 0, 1}, ImageSize → 200]
```

Out[3]=



In[4]:=
```
house = RegionPlot3D[And[0 < x, x < 1, 0 < y, y < 1 - x, 0 < z, z < x y],
  {x, -0.01, 1}, {y, -0.01, 1}, {z, -0.01, 1 / 4}, PlotPoints → {121, 121, 121},
  PlotStyle → {Opacity[0.4]}, Mesh → False, ImageSize → 250,
  AxesLabel → {x, y, z}, AxesEdge → {{-1, -1}, {-1, -1}, {-1, -1}}]
```

Out[4]=



In[5]:=
```
Options[Plot3D]
```

Out[5]=
```
{AlignmentPoint → Center, AspectRatio → Automatic, AutomaticImageSize → False,
 Axes → True, AxesEdge → Automatic, AxesLabel → None, AxesOrigin → Automatic,
 AxesStyle → {}, Background → None, BaselinePosition → Automatic, BaseStyle → {},
 BoundaryStyle → GrayLevel[0], Boxed → True, BoxRatios → {1, 1, 0.4},
 BoxStyle → {}, ClippingStyle → Automatic, ColorFunction → Automatic,
 ColorFunctionScaling → True, ColorOutput → Automatic, ContentSelectable → Automatic,
 ControllerLinking → Automatic, ControllerMethod → Automatic, ControllerPath → Automatic,
 CoordinatesToolOptions → Automatic, DisplayFunction :→ $DisplayFunction,
 Epilog → {}, Evaluated → Automatic, EvaluationMonitor → None, Exclusions → Automatic,
 ExclusionsStyle → None, FaceGrids → None, FaceGridsStyle → {}, Filling → None,
 FillingStyle → Opacity[0.5], FormatType :→ TraditionalForm, ImageMargins → 0.,
 ImagePadding → All, ImageSize → Automatic, LabelStyle → {}, Lighting → Automatic,
 MaxRecursion → Automatic, Mesh → Automatic, MeshFunctions → {#1 &, #2 &},
 MeshShading → None, MeshStyle → Automatic, Method → Automatic, NormalsFunction → Automatic,
 PerformanceGoal :→ $PerformanceGoal, PlotLabel → None, PlotPoints → Automatic,
 PlotRange → {Full, Full, Automatic}, PlotRangePadding → Automatic,
 PlotRegion → Automatic, PlotStyle → Automatic, PreserveImageOptions → Automatic,
 Prolog → {}, RegionFunction → (True &), RotationAction → Fit, SphericalRegion → False,
 TextureCoordinateFunction → Automatic, TextureCoordinateScaling → Automatic,
 Ticks → Automatic, TicksStyle → {}, ViewAngle → Automatic, ViewCenter → Automatic,
 ViewMatrix → Automatic, ViewPoint → {1.3, -2.4, 2.}, ViewRange → All,
 ViewVector → Automatic, ViewVertical → {0, 0, 1}, WorkingPrecision → MachinePrecision}
```

In[6]:=
```
Show[house]
```

Out[6]=



In[7]:=
```
Graphics[{Blue, Polygon[{{1, 2}, {3, 5}, {-3, 4}, {1, 2}}]},
 Frame → True, PlotRange → {{-5, 5}, {-5, 5}}]
```

Out[7]=



In[8]:=
```
x0 = .25; Show[house, Graphics3D[
  {Blue, Polygon[{{x0, 0, 0}, {x0, 1 - x0, 0}, {x0, 1 - x0, x0 (1 - x0)}, {x0, 0, 0}}]}]]
```

Out[8]=



In[9]:=
```
Manipulate[Show[house, Graphics3D[{Blue, Polygon[
      {{x0, 0, 0}, {x0, 1 - x0, 0}, {x0, 1 - x0, x0 (1 - x0)}, {x0, 0, 0}}]}]], {{x0, 0.5}, 0, 1}]
```

Out[9]=



A slight change in the roof

In[10]:=
```
Plot3D[Sqrt[x y], {x, 0, 1}, {y, 0, 1}]
```

Out[10]=



In[11]:=
```
houseS = RegionPlot3D[And[0 < x, x < 1, 0 < y, y < 1 - x, 0 < z, z < Sqrt[x y]],
  {x, -0.01, 1}, {y, -0.01, 1}, {z, -0.01, 1 / 2}, PlotPoints → {151, 151, 151},
  PlotStyle → {Opacity[0.4]}, Mesh → False, ImageSize → 800, AxesLabel → {x, y, z},
  AxesEdge → {{-1, -1}, {-1, -1}, {-1, -1}}, BoxRatios → {1, 1, 1 / 2}]
```

Out[11]=



In[12]:=
```
Show[houseS, ImageSize → 250]
```

Out[12]=



In[13]:=
```
x0 = .5; Show[houseS, Graphics3D[{Blue, Polygon[
    {{x0, 0, 0}, {x0, 1 - x0, 0}, {x0, 1 - x0, √(x0 (1 - x0))}, {x0, 0, 0}}]}], ImageSize → 250]
```

Out[13]=



It is more work to get this accurately.

In[14]:=
```
Graphics[{Yellow, Polygon[Table[{Cos[t], Sin[t]}, {t, 0, Pi/2, 2 Pi/150}]]},
 PlotRange → {{-2, 2}, {-2, 2}}, Frame → True, ImageSize → 200]
```

Out[14]=



In[15]:=
```
Graphics[{Yellow, Polygon[Append[Table[{Cos[t], Sin[t]}, {t, 0, Pi/2, 2 Pi/150}], {0, 0}]]},
 PlotRange → {{-2, 2}, {-2, 2}}, Frame → True, ImageSize → 200]
```

Out[15]=



In[16]:=
```
x0 = .5; Show[houseS, Graphics3D[{Blue, Polygon[
  Append[Table[{x0, y, Sqrt[x0 y]}, {y, 0, 1 - x0, 1/75}], {x0, 1 - x0, 0}]]}], ImageSize → 250]
```

Out[16]=



In[17]:=
```
Manipulate[Show[houseS, Graphics3D[
  {Blue, Polygon[Append[Table[{x0, y, Sqrt[x0 y]}, {y, 0, 1 - x0, 1/75}], {x0, 1 - x0, 0}]]}],
 ImageSize → 250], {{x0, .5}, 0, 1}]
```

Out[17]=



Making an animation

It turns out that it is easier to work with the graph of the roof if we make it using the command ParametricPlot3D. In the command bellow I also wanted to emphasise the foundation. I did it using the command Graphics3D. I combined two command using Show.

```
houseS1 = Show[ParametricPlot3D[{x, t (1 - x), Sqrt[x t (1 - x)]}, {x, 0, 1},
    {t, 0, 1}, Mesh → False, PlotStyle → {Opacity[0.4]}, PlotPoints → {51, 51}],
  Graphics3D[{Line[{{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 0}}]}],
  PlotRange → {{0, 1}, {0, 1}, {0, 1 / 2}}, ImageSize → 800]
```

Out[18]=



In[19]:=
```
Slices = Table[Show[houseS1, Graphics3D[
    {Blue, Polygon[Append[Table[{x0, y, Sqrt[x0 y]}, {y, 0, 1 - x0, 1/75}], {x0, 1 - x0, 0}]]}],
  ImageSize → 800], {x0, 0, 1, 1/50}];
```

In[23]:=
```
Show[Slices[[26]]]
```

Out[23]=



In[45]:=
```
Length[Slices]
```

Out[45]= 51

In[24]:=
```
NotebookDirectory[]
```

Out[24]=
```
C:\Dropbox\Work\myweb\Courses\225_201720\
```

Above I produced the object Slices. It is a list of 51 pictures. I can Export that list as an animated gif file. I do it in the command below. In fact I export the 26th picture in Slices separately to use it as a picture on the class website.

```
(* SetDirectory[NotebookDirectory[]];

Export["SlicesD.gif",Slices[[26]],"GIF","ImageSize"→800];

Export["SlicesAni.gif",Slices,"GIF",
 "AnimationRepetitions"→Infinity,"ImageSize"→800,"DisplayDurations"→0.3]
*)
```

Out[27]=
```
SlicesAni.gif
```

It turns out that to calculate the double integral of $\sqrt{x\,y}$ over the region $\{(x,\ y)\ :\ 0 \le x \le 1,\ 0 \le y \le 1 - x\}$ is little bit more difficult. The first integral is not difficult, please calculate it yourself.

In[28]:=
```
FullSimplify[Integrate[Sqrt[x y], {y, 0, 1 - x}], And[x > 0, x < 1]]
```

Out[28]=
$$\frac{2}{3}\,(1-x)^{3/2}\,\sqrt{x}$$

The second integral is more difficult:

In[29]:=
```
FullSimplify[Integrate[(1 - x) Sqrt[1 - x] Sqrt[x], x], And[x > 0, x < 1]]
```

Out[29]=
$$-\frac{1}{24}\sqrt{-(-1+x)\,x}\,(3+2\,x\,(-7+4\,x))+\frac{\text{ArcSin}\left[\sqrt{x}\right]}{8}$$

I will show how to do this integral by hand on the website.